

# HT56R2x 使用 SPI 進行資料傳輸之用法與 注意事項與 Software


文件編碼：HA0213T

## 簡介

HT56R2x 內建有 Serial Interface Function，其中包括了 SPI 和 I2C 這兩種總線模式，以及 SPI1 單獨資料傳輸模式，SPI1 主要用於音頻信號資料通信的傳輸，由於 HT56R2x 系列的 SPI 與 SPI1 的通信方式和控制方式類似，所以本文以 HT56R24 為母體，主要介紹使用 SPI (SIM) 進行資料傳輸的方法和注意事項。


## SPI Interface

SPI(Serial Peripheral Interface) 是一個全雙工串列資料傳輸器，最初由摩托羅拉設計，其允許多種設備通過 SPI 總線進行相互通信。設備之間通過主/從技術，只有主機能夠發起資料的傳遞。一個簡單的四線信號總線被用來進行所有的通信，並且這些引腳與普通的 I/O 口共用引腳。




SPI Block Diagram


SPI 介面是一個全雙工的串列資料連接，串列介面功能有 4 根基本信號線，包含 SDO (串列資料輸出)、SDI (串列資料輸入)、SCK (串列時鐘) 和  $\overline{SCS}$  (從元件選擇)。注意的是，從機選擇線的條件是由 SIMCTL2 控制暫存器內的 CSEN 位元決定的。如果 CSEN 被置位， $\overline{SCS}$  線有效，但如果 CSEN 被清除為 0，那麼  $\overline{SCS}$  線將處於浮空狀態。下面的時序圖描述了 SPI 總線的主模式和從模式下的時序協定。



**SPI Master Mode Timing**



**SPI Slave Mode Timing – CKEG=0**




Note: For SPI slave mode, if SIMEN=1 and CSEN=0, SPI is always enabled and ignore the  $\overline{SCS}$  level.

**SPI Slave Mode Timing – CKEG=1**

## Operation Declaration

如下圖所示，SPI 的介面與 I/O 功能和 I2C 以及 COM 通道、PWM 等共用腳位，要打開 SPI 功能你必須首先將配置選項里的 SIM Function 選擇為 Enable，WCOL 選擇為 Enable，CSEN 選擇為 Enable，然後設置好 SIMCTL0 和 SIMCTL2 暫存器值，如果在 IDLE Mode 中工作使能，則在 CLKMOD bit 4 SIMIDLE 設置為 1，則 SPI 通信工作模式在 IDLE Mode 使能，即為 Enable。而 CLKMOD bit 4 SIMIDLE 只有在 IDLE Mode 時提供給 SPI 介面 CLOCK 使用！（同時，SPI1 也支持在 IDLE Mode 下工作，也是由 CLKMOD bit 4 SIMIDLE 來決定，如果工作使能，則設置為 1，反之，設置為 0。）



HT56R24 中與 SPI 功能相關的一共有 3 個暫存器。它們是 SIMDR、SIMCTL0、SIMCTL2。SIMDR 暫存器被用來來存儲馬上要傳輸或者剛接收到的資料。它是和 I2C 共用的，在 HT56R24 中，SPI 與 I2C 只能二選其一。要把資料寫入 SPI 總線，資料必須要放入 SIMDR 暫存器才能被得到傳輸。相應的在資料從 SPI 總線接收到後，也只能從 SIMDR 暫存器中讀出資料。總之，任何通過 SPI 總線進行傳送和接受的資料都必須通過相應接收或者發送資料暫存器。

	Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
SIMDR	SD7	SD6	SD5	SD4	SD3	SD2	SD1	SD0
R/W	R/W	R/W	R/W	R/W	R/W	R/W	R/W	R/W

SIMDR 暫存器

### 資料傳輸工作模式

SIMCTL0 暫存器也是 SPI 和 I<sup>2</sup>C 共用的，它被用來打開或者關閉串列介面功能，設定 SPI 總線資料傳輸的時鐘頻率。

	Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
Label	SIM2	SIM1	SIM0	PCKEN	PCKPSC1	PCKPSC0	SIMEN	—
R/W	R/W	R/W	R/W	R/W	R/W	R/W	R/W	R

- SIMEN

這一個 Bit 在總體上控制 SPI 介面的開與關，當 SIMEN 位元被清除為 0 時，將關閉掉 SPI 介面功能。SDI、SDO、SCK 與  $\overline{\text{SCS}}$  將會處於浮空狀態，此時，SPI 工作電流將變為一個極小值。當這一位元變為“1”時，SPI 功能將打開。當然，必須要先在配置選項中打開 SIM

功能，這一位元才會有效。要注意到地方是，當 SIMEN 位元從 0 變為 1 時，此時 SPI 的控制暫存器將會變為無法預測的值，因此，每當 SIMEN 位元從 0 變為 1 時，應用程式必須重新初始化 SPI 的控制暫存器。

- PCKEN、PCKPSC1、PCKPSC0

PCKEN、PCKPSC1、PCKPSC0 這三個暫存器用於控制週邊時鐘輸出。此週邊設備時鐘輸出引腳 PCLK 與 I/O 引腳 (PB.2) 共用，這個週邊設備的時鐘源來自於定時/計數器 2 的 2 除頻或內部系統時鐘的除頻。

PCKEN：這一個位元在總體上控制週邊時鐘輸出的開與關。

PCKPSC0、PCKPSC1：選擇週邊時鐘輸出的時鐘來源，如下表：

PCKPSC [1:0]: Peripheral Clock Output Clock Source
00: $f_{SYS}$
01: $f_{SYS}/4$
10: $f_{SYS}/8$
11: Timer 2 output/2

- SIM0~SIM2

這幾位元用來設定 SIM 功能的操作模式，也就是選擇 I<sup>2</sup>C 功能或者選擇 SPI 功能。如果選擇了 SPI 功能，還將選擇主/從模式、主機的時鐘頻率，時鐘可以選擇為系統時鐘，也可以來源於計時器。如果選擇從模式，將不用選擇時鐘，它的時鐘由它的主機提供。具體情況由下圖所示：

**SIM2, SIM1, SIM0: SIM Operating Mode Control**  
 000: SPI master mode; SPI clock is  $f_{SYS}/4$   
 001: SPI master mode; SPI clock is  $f_{SYS}/16$   
 010: SPI master mode; SPI clock is  $f_{SYS}/64$   
 011: SPI master mode; SPI clock is  $f_{TBC}$   
 100: SPI master mode; SPI clock is TM0 CCRP match frequency/2  
 101: SPI slave mode  
 110: I<sup>2</sup>C slave mode  
 111: Unused mode

SIMCTL2 暫存器是一個 SPI 專用的控制暫存器。

#### SIMCTL2 Register

	Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
Label	—	—	CKPOL	CKEG	MLS	CSEN	WCOL	TRF
R/W	R/W	R/W	R/W	R/W	R/W	R/W	R/W	R/W


- TRF  
傳送/接收完成旗標位元，當傳送或者接收完成後，它將自動設定為 1。需要軟體清除。
- WCOL  
主/從模式下，若正在發送資料或接收資料，寫 SIMDR 暫存器將會設定位元 WCOL，且寫入資料被忽略。WCOL 功能可由配置選項打開或者關閉。WCOL 由硬體設定位元，軟體清除為 0。
- CSEN  
串列介面片選功能打開/關閉。若 CSEN=1， $\overline{SCS}$  片選功能有效。主模式下，在 SCK 信號輸出前輸出  $\overline{SCS}$  片選信號；而在從模式下，接收到  $\overline{SCS}$  信號前(後)，資料傳輸被關閉(打開)。若 CSEN=0， $\overline{SCS}$  引腳處於浮空狀態，選片功能失效，此時可在外部對  $\overline{SCS}$  引腳 Pull-High 以實現片選功能。CSEN 功能可由配置選項打開或者關閉。
- MLS  
MSB 或 LSB 選擇位元。也就是傳輸時高位元優先或者低位元優先。
- CKEG 和 CKPOL  
這兩位元在資料傳輸前必須進行設定，否則，一個錯誤的時鐘緣會被產生。CKPOL 位元


決定時鐘線的基本狀態，當這一位元為高位元時，時鐘線在不活動的狀態下將為低位元，反之為高位元。CKEG 位元決定時鐘線有效時的時鐘緣類型，它將依賴於 CKPOL 的值，下面是這 2 位元組合起來的情況。

CKPOL	CKEG	SCKClock Signal
0	0	High Base Level Active Rising Edge
0	1	High Base Level Active Falling Edge
1	0	Low Base Level Active Falling Edge
1	1	Low Base Level Active Rising Edge

### S/W 流程圖

以 SPI 資料傳輸為例，以下為 SPI(SIM)程序的流程圖 (SPI1 的資料傳輸方式流程圖與之類似)，如下：





SPI Transfer Control Flowchart

## SPI 操作

在與 I<sup>2</sup>C 共用 SIMCTL0 暫存器裏，在主/從機模式下，所有的通信都通過使用 SPI 匯流排得以實現。時序圖顯示了基本的總線操作。SPI 傳輸過程中，主機在輸出 SCK 信號前，通過傳輸一個  $\overline{\text{SCS}}$  信號選擇從機，當 CSEN=0， $\overline{\text{SCS}}$  引腳需外部 Pull-High，如 SPI Master Mode Timing 圖所示。從模式在  $\overline{\text{SCS}}=1$  時，SDO 為浮空， $\overline{\text{SCS}}=0$  時，SDO 有效。從模式下，若 CSEN=0，不管  $\overline{\text{SCS}}$  狀態如何（為高準位或為低準位），只要 SIMEN=1，SPI 就處於有效狀態。SIMCTL0 暫存器的 SIMEN 位元必須置位，其可以將 SDI 引腳設成浮空狀態並且將 SDO 引腳置高。在從模式下，SCK 引腳將處於浮空狀態。如果 SIMEN 位元清除為 0，那麼總線將被禁止，並且  $\overline{\text{SCS}}$ 、SDI、SDO 和 SCK 將全都處於浮空狀態。在主模式下，主機將始終產生時鐘信號。在資料被寫入 SIMDR 暫存器後，將啟動時鐘和資料的傳遞。在從模式下，資料的傳遞和接受將通過接收來自外部主機設備的時鐘信號來啟動。下面的步驟顯示了在主/從模式下資料傳遞遵循的順序。

### 主模式

- 步驟 1  
程序初始化，設定 SIMCTL0 控制暫存器的 SIM0~SIM2 位元來選擇主模式和串列傳輸速率。
- 步驟 2  
設定 CSEN 並利用 MLS 來選擇資料從高位元還是低位元開始，從機必須與主機保持一致。
- 步驟 3  
設定 SIMCTL0 控制暫存器的 SIMEN 位元來打開 SPI。
- 步驟 4  
將欲傳出資料寫入 SIMDR，檢查 WCOL：WCOL=1 → 發生衝突錯誤，並跳至步驟 4。  
WCOL=0 → 跳至步驟 5。
- 步驟 5  
檢查 TRF 或等待 SPI 串列總線中斷。
- 步驟 6  
從 SIMDR 暫存器讀取資料，放入 temp3 資料暫存器。
- 步驟 7  
清除 TRF，返回步驟 4。

### 從模式

- 步驟 1  
設定的 SIM0~SIM2 位元為“101”來選擇從模式。
- 步驟 2  
設定 CSEN 並利用 MLS 來選擇資料從高位元還是低位元開始，主機必須保持一致。
- 步驟 3  
設定 SIMCTL0 控制暫存器的 SIMEN 位元來打開 SPI 介面。
- 步驟 4  
將資料寫入 SIMDR，檢查 WCOL：WCOL=1 → 發生衝突錯誤，並跳至步驟 4。  
WCOL=0 → 跳至步驟 5。
- 步驟 5  
檢查 TRF 或等待 SPI 串列總線中斷。
- 步驟 6  
從 SIMDR 暫存器讀取資料，放入 temp3 資料暫存器。
- 步驟 7  
清除 TRF，返回步驟 4。

### SPI 配置選項

一些配置選項必須通過設備的程式來設定後用作 SPI 介面功能。一個配置選項用來在 SIMCTL2 暫存器使能 WCOL 的操作和寫入衝突位元。另一個配置選項用來禁止或使能 SIMCTL0 暫存器中的 CSEN 位元。如果配置選項禁止 CSEN 位元，那麼 CSEN 將不能用來影響 SPI 匯流排所有的控制。SIMCTL2 暫存器的 WCOL 位元用來在資料傳送中提示寫入衝突錯誤。當資料傳遞操作出現寫 SIMDR 暫存器的現象時，WCOL 位元會提示資料衝突並且防止繼續進行寫操作。WCOL 位元將通過硬體被置位，但必須通過用戶應用程式來清除為 0。WCOL 位元的全部功能可以通過配置選項來禁止或打開。

### 對應的程式注意事項

在設備進入 IDLE 模式下，注意的是，資料的接收和傳遞將依賴於 FSYSON 位元，它位於 WDTC 暫存器的第 7 位元，當 FSYSON 位元為“1”時，方可以在 IDLE 模式下繼續進行資料的接收和傳遞。在資料傳遞和接收後，TRF 位元用來產生中斷 (SPI1 的程式注意事項與之類似)。

## 程式說明

本範例由主程式以及 SPI 的服務程式二部分組成，在主程式中，會對 MCU 進行初始化操作，主要是對 SPI 的相關暫存器進行初始化操作，關閉一些和 SPI 功能共用 I/O 的功能。

SPI 的服務程式 spi\_Transmission 副程式中，會將資料暫存器 temp1 中的資料發送到 SPI 總線，會將從 SPI 總線上接收到的資料保存到 temp3 中，用戶可以自行修改。如果用戶只需要發送或者接收，就只需要注意 temp1 或者 temp3 的值。

程式中沒有打開中斷，如果要使用中斷，請在檢查 WOCL 為“0”後，不必再繼續檢查 TRF 位元，打開相應中斷使能位元 ESIM，總中斷使能位元 EMI，等待中斷產生。

配置選項中，要使用 SPI 總線，必須打開 SIM Function，WOCL 和 CSEN。其他選項由用戶使用情況決定。附件 HT56R24(SPI SIM).zip 為 HT56R24 在 SPI 工作模式下與 HT66F40 的電路原理圖。

## 程式範例

在 SPI(SIM)工作模式下的主/從機發送/接收程式：

```

;function:edit for SPI master
;MCU HT56R24
;option:
;OSC HIRC 4MHz
;SIM enable
;WCOL enable
;CSEN enable
;IO or resb IO
;WDT disable
;SPI1 disable
;SPI1 WCOL disable
;SPI1 CSEN disable
;pa5/6/7 VDD
;VDDIO disable
;pb4/5 IO
;pc6 IO
;others select by user
include HT56R24.inc
ds .section 'data'
temp1 db ?
temp2 db ?
temp3 db ?
temp4 db ?
temp5 db ?
temp6 db ?
temp7 db ?
temp8 db ?
cs .section 'code'
org 000h
jmp main
org 014h
reti
mmov macro h2,h1
mov a,h1
mov h2,a
endm
main:
    call BK0
    call BK2
    call BK3
    call BK4
    call BK5 ;clr data of the bank0~5,except bank1
    set pec2 ;set pe2 as input
    set pepu2
    set pec3 ;set pe3 as input
    set pepu3
    set pec4 ;set pe4 as input
    set pepu4
    clr emi
    clr tmr1c
    clr intc0
    clr intc1
    clr simdr
    clr temp1
    clr temp2
    clr temp3
    
```

```

        clr     temp4
        clr     temp5
L0:sz    pe2           ;choose the transmit mode of SPI
        jmp    L1           ;pe2=1,slave
        jmp    L2           ;pe2=0,master,fsys/4
L1:clr   csen
        mov    a,0a0h
        jmp    L3
L2:set   csen
        mov    a,000h
L3:mov   simctl0,a
        clr   ckpol           ;high Base Level active rising edge
        clr   ckeg
        set   mls           ;data shift oder from msb to lsb
        set   simen         ;enable sim
        jmp   spi_Transmission
L:  sz    pe3
        jmp   Linput         ;pe3=1, pa port as input mode
        jmp   Loutput        ;pe3=0, pa port as output mode
Linput:
        ;set the pa port input mode one time in order not to inflict the
        transmission
        mov   a,temp4
        addm a,pcl
        jmp   Linput1
        jmp   Linput2
Linput1:
        set   pac           ;set pa port as input
        set   papu
Linput2:
        mmov temp1,pa         ;mov pa to temp1 register
        mmov temp4,1
        clr   temp5
        jmp   spi_Transmission
Loutput:
        mov   a,temp5
        addm a,pcl
        jmp   Loutput1
        jmp   Loutput2
Loutput1:
        clr   pac           ;set pa port as output
        set   pa
Loutput2:
        mmov pa,temp3         ;mov the simdr data into temp3 and display by pa port
        mmov temp5,1
        clr   temp4
        jmp   spi_Transmission
spi_Transmission:           ;the transmission of SPI programme
        set   simen
        clr   trf
        clr   wcol
        mmov simdr,temp1     ;mov the register data of temp1 into simdr,ready to transmit
        sz    wcol
        jmp   $-4
        snz   trf
        jmp   $-1
        mmov temp3,simdr
        snz   pe4
        jmp   L0           ;pe4=0,return to judge the state of pe2,and judge the transmit
slave or master mode
        jmp   L           ;pe4=1,return to judge the state of pe3,and judge the pd input
or output mode

```

備註：

- pe2, pe3:
  - 00: 作為主機，輸出，可以通過主機看到從機發送而被主機接收到的資料。
  - 10: 作為主機，輸入，可以通過從機看到主機發送而被從機接收到的資料。
  - 01: 作為從機，輸出，可以通過從機看到主機發送而被從機接收到的資料。
  - 11: 作為從機，輸入，可以通過主機看到從機發送而被主機接收到的資料。
- pe4 :
  - 0: 重新檢測 pe2 埠高/低的狀態，重新設定在 SPI 通信時的主/從模式。
  - 1: 進入檢測 pe3 埠高/低的狀態，設定所 SIMDR 暫存器資料的存入/取出方式。

與 SPI 工作模式下相對應的從/主機接收/發送程式：

```

;function: edit for SPI master
;MCU: HT66F40
;option:
;OSC  $\Delta$  HIRC + 4M + LIRC 32K
;SIM enable
;WCOL enable
;CSEN enable
;I/O or RESB  $\rightarrow$  RESB
;WDT Enable two instructions
;Others select by user
include HT66F40.INC
.listinclude
use_data .section 'data'
    data1 db      ?
    data2 db      ?
    data3 db      ?
    temp1 db      ?
    temp2 db      ?
maincode .section at 00 'code'
    org      00h
    jmp     main_start
mmov     macro h2,h1
mov      a,h1
mov      h2,a
endm
main_start:
    call    initial      ;initial the program
    set     pac
; the transmission of SPI programme
spi_server:
    set     csen          ;SPI SCS pin Control--enable
    set     mls           ;SPI Data shift order---MSB
    sz     pa.0
    jmp     L1
    jmp     L2
L1:mov    a,0a0h          ;pa.0=1 spi slave mode
    jmp     L3
L2:mov    a,020h          ;pa.0=0 spi master mode
L3:mov    simc0,a
    set     simen
    jmp     spi_Transmission
L: snz    pa.1
    jmp     pc_as_output  ;pa1=0,pc port as output
    jmp     pc_as_input   ;pa1=1,pc purt as input
pc_as_output:
    mov     a,temp1
    addm   a,pcl
    
```

```

        jmp     Loutput1
        jmp     Loutput2
Loutput1:
        clr     pcc
        clr     pc
Loutput2:
        mmov pc,data2
        mmov temp1,1
        clr     temp2
        jmp     spi_Transmission
pc_as_input:
        mov     a,temp2
        addm a,pc1
        jmp     Linput1
        jmp     Linput2
Linput1:
        set     pcc
Linput2:
        mmov data1,pc
        mmov temp2,1
        clr     temp1
        jmp     spi_Transmission
spi_Transmission:
        clr     trf
        clr     wcol
        mov     a,data1
        mov     simd,a           ;move data1 to simd register
        sz     wcol             ;judge the mode of write whether or not conflict
        jmp     $-4
        snz    trf              ;judge the transmission finish
        jmp     $-1
        mmov data2,simd        ;read data from simd
        snz    pa2
        jmp     spi_server      ;pa2=0, return to judge the state of pa1 port
        jmp     L               ;pa2=1, return to judge the state of pa0 port
    
```

備註：

- pa0、pa1:
  - 00: 作為主機，輸出，可以通過主機看到從機發送而被主機接收到的資料。
  - 10: 作為主機，輸入，可以通過從機看到主機發送而被從機接收到的資料。
  - 01: 作為從機，輸出，可以通過從機看到主機發送而被從機接收到的資料。
  - 11: 作為從機，輸入，可以通過主機看到從機發送而被主機接收到的資料。
- pa2:
  - 0: 重新檢測 pa0 埠高/低的狀態，並重新設定在 SPI 通信時的主/從模式。
  - 1: 檢測 pa1 埠高/低的狀態，設定所 SIMD 暫存器資料的存入/輸出方式。

## 總結

本文主要講解了 HT56R24 的 SPI 操作與注意事項，由於 SPI1 的操作與注意事項類似與 SPI，所以用戶若需要使用 SPI1，則可以對相應的暫存器進行設定，如下：

SPI(SIM)	SIMCTL0	SIMCTL2	SIMDR
SPI1(Voice)	SPICTL0	SPICTL1	SPIDR

在暫存器 SPICTL0 中只有設定主機、頻率或從機，無設定 I<sup>2</sup>C 項。同時，在配置選項中應選擇 SPI1 為 Enable、WCOL1 為 Enable、CSEN1 為 Enable。如果要使用 SPI1 中斷，請在 WCOL1 為“0”後，不必再繼續檢查 TRF1 位元，打開相應中斷使能位元 ESPI，總中斷使能位元 EMI，等待中斷的產生。需要注意的是，SPI (SIM) 的中斷發生地址為 014H，而 SPI1 (Voice) 的中斷發生地址為 018H。其他選項均可參照上面表格對應的暫存器選項進行設定。用戶可以在自己的程序中直接插入此 IP，可根據自己的需要按照上面程式說明自行修改。附件中附有 HT56R24 與 HT66F40 的 SPI(SIM)和 SPI1(Voice)的通信方式以及電路原理圖。